

COMP 410
Spring 2018
Final Practice Exam

The topics on this practice exam reflect **ONLY** those which have been covered since the last exam. The real final is **CUMULATIVE**, so it will include questions similar to the previous practice exams. However, the final will be biased towards the sort of questions below.

Prolog Metainterpreters

1.) Write a metainterpreter which shows the number of conjunctions which were needed to compute a particular solution. Example queries follow. Your metainterpreter needs to handle only the rules necessary to execute these queries below.

```
?- interpret((X is 1 + 1, Y is 2 + 2), ConjunctionCount).  
X = 2, Y = 4, ConjunctionCount = 1.
```

```
% This definition is used in the query below  
% myLength([], 0).  
% myLength([_|T], Len) :-  
%     myLength(T, TLen),  
%     Len is TLen + 1.
```

```
?- interpret(myLength([a, b, c, d], Len), ConjunctionCount).  
Len = 4, ConjunctionCount = 4.
```

```
interpret(true, 0) :- !.  
interpret(is(A, B), 0) :-  
    !,  
    is(A, B).  
interpret((A, B), FinalCount) :-  
    !,  
    interpret(A, ACount),  
    interpret(B, BCount),  
    FinalCount is ACount + BCount + 1.  
interpret(Call, Count) :-  
    clause(Call, Body),  
    interpret(Body, Count).
```

Second-Order Procedures

2.) Consider the following procedure:

```
foo(1, 2, 3, 4).  
foo(1, 2, 4, 3).  
foo(3, 4, 1, 2).  
foo(1, 2, 1, 6).
```

With the above procedure in mind, provide answers to the following queries. You should provide solutions for all variables which have solutions (hint: this won't necessarily be all variables involved).

2.a) ?- findall(A, foo(A, B, C, D), E).

```
E = [1, 1, 3, 1].
```

2.b) ?- bagof(bar(C, D), foo(A, B, C, D), E).

```
A = 1, B = 2,  
E = [bar(3, 4), bar(4, 3), bar(1, 6)] ;  
A = 3, B = 4,  
E = [bar(1, 2)].
```

2.c) ?- bagof(bar(A, C, D), foo(A, B, C, D), E).

```
B = 2,  
E = [bar(1, 3, 4), bar(1, 4, 3), bar(1, 1, 6)] ;  
B = 4,  
E = [bar(3, 1, 2)].
```

2.d) ?- bagof(A, C^D^foo(A, B, C, D), E).

```
B = 2, E = [1, 1, 1] ;  
B = 4, E = [3].
```

2.e) ?- setof(A, B^D^foo(A, B, C, D), E).

```
C = 1, E = [1, 3] ;  
C = 3, E = [1] ;  
C = 4, E = [1].
```

Mercury

3.) Write a procedure named `sublist` which will take a list and nondeterministically produce lists which contain elements in the input list. Example queries follow:

```
?- sublist([], List).
List = [].

?- sublist([1], List).
List = [1] ;
List = [].

?- sublist([1, 2], List).
List = [1, 2] ;
List = [1] ;
List = [2] ;
List = [].
```

Be sure to write appropriate `pred` and `mode` annotations. You only need to write `mode` annotations corresponding to the queries above.

```
:- pred sublist(list(A), list(A)).
% :- pred sublist(list(int), list(int)). % also acceptable
:- mode sublist(in, out) is multi.
sublist([], []).
sublist([H|T], [H|Rest]) :-
    sublist(T, Rest).
sublist([_|T], Rest) :-
    sublist(T, Rest).
```

4.) Write a procedure that conforms to the following `pred` and `mode` annotations:

```
:- pred foo(int, int).
:- mode foo(in, out) is multi.
:- mode foo(in, in) is semidet.

foo(1, 2).
foo(_, 3).
```

5.) Consider the following Mercury code, which does not compile as written:

```
:- type my_type(A) ---> foo(A) ; bar(A, A) ; baz(A, A, A).  
  
:- pred something(my_type(A), A).  
:- mode something(in, out) is det.  
something(foo(A), A).  
something(bar(A, _), A).
```

5.a) Why doesn't this code compile?

The `mode` annotation claims that this code is deterministic for `(in, out)`, but currently it's only semideterministic for `(in, out)`. The code is missing a case for `baz`.

5.b) It is possible to get this code to compile by changing the `type` definition. Write a revised `type` definition below which will allow this code to compile. You may assume that this code is in complete isolation, so changing the `type` definition won't break anything elsewhere.

```
:- type my_type(A) ---> foo(A) ; bar(A, A).
```

5.c) It is possible to get this code to compile by changing the `mode` annotation. Write a revised `mode` annotation below which will allow this code to compile. As before, you may assume the code is in complete isolation.

```
:- mode something(in, out) is semidet.
```

5.d) It is possible to get this code to compile by adding another line of code at the end. Write this added line below. This line may do whatever it wants, as long as it allows the code to compile. As before, you may assume the code is in complete isolation.

```
something(baz(A, _, _), A).
```